# Power Transfer in Subsea Cables

P. McGill
29Oct12

## Introduction

Selecting electrical cable for seafloor observatories can be a daunting task that must meet multiple, conflicting requirements. Typically you want to maximize the power delivered to the load over the required distance while minimizing the weight and the cost of the cable. Further complicating matters, electrical cable is constructed from a discrete number of conductors of standard wire sizes. It's less effective to give cable manufacturers and vendors your high-level engineering requirements than it is to simply tell them how many conductors of what size you desire.

Thus, specifying electrical cable is an iterative process of selecting a candidate cable, calculating the resistance of its conductors based on the required cable length and a table of standard wire characteristics, combining multiple conductors in parallel when possible, and finally analyzing a circuit with the cable as a resistive element to see if the voltage at the load is sufficient at the required power levels. If the cable doesn't sastify the requirements then a cable with larger wire or more conductors must be evaluated. If the cable exceeds the requirements then the designer has to wonder if a smaller, less expensive cable would suffice.

What is needed is a quick way to compare various cable designs. In this paper I derive a cable characteristic called *wattacity* (pronounced wah-TA-city, a portmanteau of *watt* and *capacity*), which defines a cable's ability to deliver power over a unit distance. This is analogous to the already-established term *ampacity*, which defines a cable's ability to carry current. Wattacity is defined with units of power times length. For example, a cable with a wattacity of 100 W-km can transmit 100 W of power over a distance of 1 km, or 50 W over 2 km, or 20 W over 5 km, etc. Given a cable's wattacity, it's a simple matter to see if it will serve in a particular application.

All calculations are done here with *Mathematica*, where input expressions are followed immediately by their calculated result.

```
2 + 2
```

```
4
```

## Ideal Analysis

This analysis assumes a DC power system with no reactive effects, as is common with seafloor instrumentation. Thus we have a DC power source at voltage $V_S$ feeding a load of resistance $R_L$ through a cable with a one-way resistance of $R_C$.




We start by defining the power at the load (note I write $P_L$ as capital P followed by lower-case L).

```
Clear[Pl, Vl, Rl, Rc, Vs, ρ]
```

```
Pl = Vl^2/Rl
```

$$\frac{Vl^2}{Rl}$$

The voltage at the load can be defined in terms of the source voltage and circuit resistances using the voltage divider rule.

```
Vl = Vs Rl/(2 Rc + Rl)
```

$$\frac{Rl\,Vs}{2\,Rc + Rl}$$

Now we can see $P_L$ defined in terms of the source voltage and circuit resistances.

```
Pl
```

$$\frac{Rl\,Vs^2}{(2\,Rc + Rl)^2}$$

To deliver maximum power to the load for a given cable, we know that the load resistance must equal the total cable resistance (see [Cartwright 2008] for an example of this derivation). We can define the maximum power under the condition that the source and load resistances are the same. (Note: the following equation means "define $P_{\text{Lmax}}$ as $P_L$ given the condition that $R_L = 2 \times R_C$")

```
Plmax = Pl /. Rl → 2 Rc
```

$$\frac{\text{Vs}^2}{8\ \text{Rc}}$$

Working with the cable resistance is inconvenient because we specify cable by wire size and number of conductors. We can define $R_C$ in terms of the cable's more fundamental characteristics.

$$\text{Rc} = \frac{\rho\ \text{L}}{\text{A}}$$

$$\frac{\text{L}\ \rho}{\text{A}}$$

where $\rho$ is the resistivity of the conductor (typically copper), $L$ is the length of the cable, and $A$ is the total cross-sectional area of the conductors.

Now the maximum power delivered is

```
Plmax
```

$$\frac{\text{A}\ \text{Vs}^2}{8\ \text{L}\ \rho}$$

If we multiply $P_{\text{Lmax}}$ by the length $L$, we can get the wattacity of the cable defined in terms of the source voltage and cable characteristics.

```
Wc = Plmax L
```

$$\frac{\text{A}\ \text{Vs}^2}{8\ \rho}$$

We can define a function that calculates the cross-sectional area of a wire in square meters from its American Wire Gauge (AWG) number [Wikipedia 2012].

$$\text{Ac[awg\_]} = 1.2668 \times 10^{-8}\ 92^{\frac{(36-\text{awg})}{19.5}}$$

$$1.2668 \times 10^{-8} \times 92^{0.0512821\ (36-\text{awg})}$$

Test the function by calculating the cross-sectional area of a 20 AWG wire in square millimeters.

$$\text{Ac[20]}\ 10^6$$

```
0.517632
```

This matches the value in standard wire tables. Note that a bundle of N conductors would have N times the area of a single conductor, and $\frac{1}{N}$ times the single conductor's resistance.

Now define the resistivity of copper at 20 °C. This is a conservative figure for deep seafloor cables where the temperature is closer to 2 °C.

```
ρ = 1.68 × 10^-8
```

```
1.68 × 10^-8
```

What is the resistance per 1000 ft of a single 26 AWG copper conductor? (note the conversion from feet to meters)

$$\frac{\rho\ 1000 \times 0.3048}{\text{Ac}[26]}$$

```
39.7691
```

This is close to the wire table value of 40.81 Ohms, the difference most likely being in the value of $\rho$.

Finally we can write a function that calculates the wattacity of a cable in Watt-kilometers at source voltage $V_s$ with $Nc$ conductors each way of gauge *awg*.

```
Clear[Wc]
```

$$\text{Wc}[\text{Vs\_}, \text{Nc\_}, \text{awg\_}] = \frac{\text{Vs}^2\ \text{Nc}\ \text{Ac}[\text{awg}]}{8\ \rho\ 1000}$$

$$0.000094256 \times 92^{0.0512821\ (36-\text{awg})}\ \text{Nc}\ \text{Vs}^2$$

What is the wattacity in Watt-kilometers of South Bay Cable SB-46974A, containing six conductors in each direction each of size 26 AWG, driven at the rated voltage of 600 VDC?

```
Wc[600, 6, 26]
```

```
2069.35
```

How many Watts of power can be transmitted over a 3410-m length of this cable?

```
Wc[600, 6, 26] / 3.41
```

```
606.846
```

We can create a table showing the wattacity of cable with 1 to 6 conductors each way of gauge 26 to 16 AWG, driven at 600 VDC.

```
Labeled[Framed[TableForm[Table[Round[Wc[600, N, n]], {n, 26, 16, -2}, {N, 1, 6}],
   TableHeadings → {Range[26, 16, -2], Range[6]}]],
 {AWG, "Number of Conductors Each Way", "Cable Wattacity (W-km) at 600 VDC"},
 {Left, Top, Bottom}]
```

Number of Conductors Each Way

| AWG | 1 | 2 | 3 | 4 | 5 | 6 |
|---|---|---|---|---|---|---|
| 26 | 345 | 690 | 1035 | 1380 | 1724 | 2069 |
| 24 | 548 | 1097 | 1645 | 2194 | 2742 | 3290 |
| 22 | 872 | 1744 | 2616 | 3488 | 4360 | 5232 |
| 20 | 1387 | 2773 | 4160 | 5546 | 6933 | 8319 |
| 18 | 2205 | 4409 | 6614 | 8819 | 11 023 | 13 228 |
| 16 | 3506 | 7011 | 10 517 | 14 022 | 17 528 | 21 033 |

Cable Wattacity (W-km) at 600 VDC

# Practical Analysis

The previous analysis defined the cable wattacity under somewhat unrealistic conditions. Although this defines the absolute maximum power that the cable is capable of delivering, you are unlikely to have a source voltage or load that achieves this maximum in any practical situation. For instance, it isn't wise to operate the cable with a source voltage equal to the cable's rated voltage, as insulation can break down over time and cause faults. Insulation ratings come in large, discrete steps (e.g. 300 V, 600 V, 1000 V), so the source voltage may be much less than the cable's rated maximum.

Another limitation is the load. The most common load for these systems is a DC-to-DC converter so that the voltage delivered to an instrument doesn't vary with the load current. All DC-to-DC converters have a minimum voltage below which they will cease to operate and a maximum voltage above which permanent damage will occur. Under conditions of maximum power transfer, the voltage feeding the converter drops to only one-half of the source voltage. However, operating a converter near its rated minimum voltage can cause instability and erratic operation. At a constant power output, a slight drop in the converter's input voltage will cause the converter to draw more current, causing the input voltage to drop even more and leading to a runaway situation. It is therefore unwise to operate the converter at less than 10% above its minimum rated input voltage. Conversely, when the converter is shut down or the load is removed from its output, it must tolerate the full source voltage; the converter's rated maximum input voltage must be safely above this. Most converters can tolerate an input voltage range of between 2:1 and 3:1 (e.g. 36 to 75 V is 2.08:1), and this range may be less than the load voltage swing of your cable operating at minimum and maximum power.

Thus, the wattacity of a practical subsea cable is defined by not only the cable's characteristics, but also by the characteristics of the system in which it is installed. We need to define the *system* wattacity, $W_S$, which takes into account both the available source voltage $V_S$ and the minimum voltage that can be tolerated by the load, $V_{\text{Lmin}}$. Of course if this minimum voltage is less than half of the source voltage, then the load can operate at the maximum power point of the cable and the previous analysis applies.

In the case where $V_{\text{Lmin}} > \frac{V_S}{2}$, we know that the maximum power achieved is

```
Clear[Plmax, Rl, Rc, Vlmin, Vs]
```

```
Plmax = Vlmin^2 / Rl
```

$$\frac{\text{Vlmin}^2}{\text{Rl}}$$

The voltage at the load can be defined with the voltage divider rule.

```
eqn1 = Vlmin == Vs Rl / (2 Rc + Rl)
```

$$\text{Vlmin} == \frac{\text{Rl Vs}}{2\,\text{Rc} + \text{Rl}}$$

We can then solve the equation for $R_L$.

```
soln1 = Solve[eqn1, Rl]
```

$$\left\{\left\{\text{Rl} \to -\frac{2\,\text{Rc Vlmin}}{\text{Vlmin} - \text{Vs}}\right\}\right\}$$

Substituting the solution back into the equation for $P_{\text{Lmax}}$ gives us

```
Plmax /. soln1
```

$$\left\{-\frac{\text{Vlmin}\,(\text{Vlmin} - \text{Vs})}{2\,\text{Rc}}\right\}$$

As before, we can use this result to define the cable resistance in terms of the resistivity, length, and cross-sectional area of the wires, and create a new function to calculate the system wattacity in W-km.

```
Ws[Vs_, Vlmin_, Nc_, awg_] = Vlmin (Vs - Vlmin) Nc An[awg] / (2 ρ 1000)
```

$$0.000377024 \times 92^{0.0512821\,(36-\text{awg})}\ \text{Nc Vlmin}\ (-\text{Vlmin} + \text{Vs})$$

What is the system wattacity for the South Bay Cable SB-46974A (12 conductors total, 26 AWG), using the MARS system voltage of 375 VDC and restricting the load voltage to be no less than 275 VDC?

```
Ws[375, 275, 6, 26]
```

632.3

We can check to see that this number makes sense. At this power level we know that the current in the cable is the power at the load divided by the voltage at the load.

```
Ic = Ws[375, 275, 6, 26] / 275
```

2.29927

We can find the resistance of 1 km of cable from the previous equation for wire cross-sectional area.

$$Rc = \frac{\rho\, 1000}{6\, An[26]}$$

```
21.746
```

We can find the voltage drop in the cable by multiplying the current times the cable resistance.

```
Vc = Ic 2 Rc
```

```
100.
```

This is indeed the difference between the 375 VDC source voltage and the 275 VDC load voltage.

Now we can make a table of the system wattacity for various cables on MARS, assuming the voltage at the load must stay above 275 VDC.

```
Labeled[Framed[TableForm[Table[Round[Ws[375, 275, N, n]], {n, 26, 16, -2}, {N, 1, 6}],
   TableHeadings → {Range[26, 16, -2], Range[6]}]],
 {"AWG", "Number of Conductors Each Way", "System Wattacity (W-km) at 375 VDC"},
 {Left, Top, Bottom}]
```

Number of Conductors Each Way

| AWG | 1 | 2 | 3 | 4 | 5 | 6 |
|---|---|---|---|---|---|---|
| 26 | 105 | 211 | 316 | 422 | 527 | 632 |
| 24 | 168 | 335 | 503 | 670 | 838 | 1005 |
| 22 | 266 | 533 | 799 | 1066 | 1332 | 1599 |
| 20 | 424 | 847 | 1271 | 1695 | 2118 | 2542 |
| 18 | 674 | 1347 | 2021 | 2695 | 3368 | 4042 |
| 16 | 1071 | 2142 | 3213 | 4285 | 5356 | 6427 |

System Wattacity (W-km) at 375 VDC

# Refinements

## Wire Lay Angle

When cables are manufactured, the individual wires inside the cable are twisted into a helix. Consequently, the length of the wires is greater than the length of the cable containing the wires. This effect increases as the lay angle (i.e. the amount of twist) increases. To quantify this effect, we can look at an illustration from Machinery’s Handbook [Oberg 2004].




;

This shows that the wire can be thought of as the hypotenuse of a right triangle wrapped around a cylinder with a radius equal to that of the cable. What we want to calculate is the ratio of *H*, the "Development of Helix", to *L*, the "Lead of Helix", as a function of angle *A*, the lay angle. This ratio, which I call the *lengthening factor*, represents how much longer the wire becomes by wrapping around the cylinder instead of lying directly along it (i.e. a lay angle of zero).

We can see that the tangent of angle *A* is the circumference of the cylinder divided by the lead *L*.

```
Clear[A, H, L, R]
```

```
eqn2 = Tan[A] == 2 Pi R / L
```

$$\mathrm{Tan}[A] == \frac{2\,\pi\,R}{L}$$

We can now solve this for *L*.

```
soln2 = Solve[eqn2, L]
```

$\{\{L \to 2\,\pi\,R\,\mathrm{Cot}[A]\}\}$

The length of the helix *H* can be found with the Pythagorean theorem.

```
H = Sqrt[L^2 + (2 Pi R)^2]
```

$$\sqrt{L^2 + 4\,\pi^2\,R^2}$$

Now we can write the expression for lengthening factor, substituting in the expression for *L*.

```
lf = H/L /. soln2
```

$$\left\{\frac{\sqrt{4\,\pi^2\,R^2 + 4\,\pi^2\,R^2\,\mathrm{Cot}[A]^2}\;\mathrm{Tan}[A]}{2\,\pi\,R}\right\}$$

We can simplify the expression.

```
Simplify[lf]
```

$$\left\{\frac{\sqrt{R^2\,\mathrm{Csc}[A]^2}\;\mathrm{Tan}[A]}{R}\right\}$$

We can simplify much more by restricting *R* and *Csc[A]* to be positive.

```
Simplify[lf, {R > 0, Csc[A] > 0}]
```

{Sec[A]}

The lengthening factor is simply the secant of the lay angle. Now we can plot the lengthening factor in percent of length added as a function of lay angle in degrees.

```
Plot[(Sec[A Degree] - 1) 100, {A, 0, 45},
 {Frame → True, GridLines → Automatic, FrameLabel → {{"Increase in Length (%)", ""},
    {"Lay Angle (degrees)", "Lengthening Factor vs. Lay Angle"}}}]
```




Most lay angles are in the range of 15 to 25 degrees, yielding length increases of about 4 to 10 percent.

```
Plot[(Sec[A Degree] - 1) 100, {A, 15, 25},
 {Frame → True, GridLines → Automatic, FrameLabel → {{"Increase in Length (%)", ""},
    {"Lay Angle (degrees)", "Lengthening Factor vs. Lay Angle"}}}]
```




The power conductors in South Bay Cable SB-46974A have a lay angle of 20°, and so the wires in this cable are 6.4% longer then the length of the cable itself.

```
Sec[20 Degree] // N
```

```
1.06418
```

The lengthening factor could be incorporated into the formulas for wattacity defined above, but I prefer to keep the formulas simple and simply adjust the length of the cable. For instance, if you know the application requires a 1 km cable and the lay angle is 20°, then divide the system wattacity by 1064 m to find the maximum delivered power. If you don't know the lay angle of the cable, then it's probably safe to assume a 10% lengthening factor.

## Odd Number of Conductors

If the cable you're considering has an odd number of conductors, the leftover conductor should certainly be used as it increases the wattacity of the cable. The resistance of one of the legs going to or from the load is reduced, causing the voltages at each end of the load to shift up or down together so that they're no longer distributed equally on either side of $\frac{V_S}{2}$, but this doesn't adversely affect the operation of the load's DC-DC converter. But how do you take the extra conductor into account when calculating wattacity, since the formula assumes an equal number of conductors in each direction?

You may be tempted to distribute the extra conductor evenly between the two sides by using a non-integer number for N (e.g. given five conductors, you could set N equal to 2.5). This is not a good approach as it overestimates the wattacity of the cable, especially when the total number of conductors is small. For example, if your cable has three 1 Ohm conductors and you parallel two of them together for one side of the line, then the total cable resistance is

```
actual = 1 + 1/(1 + 1)
```

$$\frac{3}{2}$$

However, if you assume that the copper in the third conductor is equally distributed between the first and second conductors (physically impossible with real wires), then this is equivalent to putting a 2 Ohm conductor in parallel with each 1 Ohm line. Now the total resistance is

```
assumed = 2 × 1/(1/1 + 1/2)
```

$$\frac{4}{3}$$

```
actual/assumed // N
```

```
1.125
```

In this case the actual resistance of the cable is over 12% more than assumed and the wattacity is correspondingly reduced.

There are at least two ways to solve this problem. The first is to modify the original wattacity formula to accept a different number of conductors for the two legs.

$$\mathbf{Ws2[Vs\_,\ Vlmin\_,\ N1\_,\ N2\_,\ awg\_]} = \frac{\mathbf{Vlmin\ (Vs - Vlmin)}}{\rho\ 1000 \left(\frac{1}{\mathbf{N1\ An[awg]}} + \frac{1}{\mathbf{N2\ An[awg]}}\right)}$$

$$\frac{59\,523.8\ \mathrm{Vlmin}\ (-\mathrm{Vlmin} + \mathrm{Vs})}{\frac{7.89391\times10^{7} \times 92^{-0.0512821\ (36-\mathrm{awg})}}{\mathrm{N1}} + \frac{7.89391\times10^{7} \times 92^{-0.0512821\ (36-\mathrm{awg})}}{\mathrm{N2}}}$$

If this formula is correct then setting the number of conductors in each direction equal should yield the same wattacity of 632.3 W-km for the South Bay Cable SB-46974A.

```
Ws2[375, 275, 6, 6, 26]
```

```
632.3
```

It's interesting to note that if we were to mis-wire the cable and place more conductors in one leg than the other, the wattacity would be reduced.

```
Ws2[375, 275, 5, 7, 26]
```

```
614.736
```

A second way to solve the odd-conductor problem is to simply round the number of conductors down to the next even integer, assume that each leg has the same number of conductors for the wattacity calculation, and treat the extra conductor as an added safety margin when it's wired into the circuit. We can make a table showing the increase in wattacity due to the additional conductor.

```
Labeled[Framed[TableForm[Table[
    Round[((Ws2[375, 275, N, N + 1, 26] / Ws[375, 275, N, 26]) - 1) 100, 0.1], {N, 1, 10}],
   {TableDirections → Row, TableHeadings → Automatic}]],
 {"Number of Conductors Each Way",
  "Increase in Wattacity Due to Additional Conductor (%)"}, {Top, Bottom}]
```

Number of Conductors Each Way

| 1 | 2 | 3 | 4 | 5 | 6 | 7 | 8 | 9 | 10 |
|---|---|---|---|---|---|---|---|---|---|
| 33.3 | 20. | 14.3 | 11.1 | 9.1 | 7.7 | 6.7 | 5.9 | 5.3 | 4.8 |

Increase in Wattacity Due to Additional Conductor (%)

As you would expect, the increase is greatest for a small number of conductors.

# Examples

## MOBB Cable

The MOBB (Monterey Ocean-Bottom Broadband) seismometer cable is South Bay Cable SB-46974A and has a total of 12 power conductors each of size 26 AWG. The power conductors are copper and have a lay angle of 20°. The MOBB observatory is connected to the MARS (Monterey Accelerated Research System) network node with 3410 m of cable. The MARS node supplies power at 375 VDC.

**If the MOBB system contains a Vicor DC-DC converter with an allowed input voltage range of 250-425 VDC, how much power can be delivered?**

Adding 10% to the minimum input voltage, the system wattacity of this cable is

```
Wsvicor = Ws[375, 275, 6, 26]
```

```
632.3
```

The actual length of the wire in the cable in km is

```
Lc = 3.410 Sec[20 Degree] // N
```

```
3.62885
```

The maximum delivered power is the wattacity divided by the length.

$$\text{Pvicor} = \frac{\text{Wsvicor}}{\text{Lc}}$$

```
174.243
```

A 300 W Vicor "mini" module would suffice for this application. Note that this is the input power to the Vicor and less power would be delivered to the actual load due to losses in the DC-DC converter.

**How much more power can be delivered if a different DC-DC converter with a lower allowable input voltage is used?**

The SynQor IQ4H line of DC-DC converters has an allowable input range of 180-475 VDC. Once again adding 10% to the minimum voltage, the system wattacity of the cable is

```
Wssynqor = Ws[375, 198, 6, 26]
```

```
805.804
```

$$\text{Psynqor} = \frac{\text{Wssynqor}}{\text{Lc}}$$

```
222.055
```

A 300 W SynQor “half brick” module would work here.

$$\frac{\text{Psynqor}}{\text{Pvicor}}$$

```
1.2744
```

The power delivered to the DC-DC converter has been increased by 27%.

**If the MOBB system requires a maximum of 40 W to operate, what are some smaller cables that would deliver this power at the same distance through a Vicor DC-DC converter?**

Since we have specified the minimum power that must be delivered to the instrumentation, we must account for the DC-DC converter’s efficiency. The efficiency of 75 W Vicor “micro” modules is specified in their data sheet only at nominal input voltage and full output load, so we’ll assume an efficiency of 60% for our application (some testing would be warranted here). The input power to the converter would be

$$\frac{40}{0.6}$$

```
66.6667
```

We can create a table that divides the system system wattacity by the adjusted length of the cable.

```
Labeled[Framed[TableForm[Table[Round[Ws[375, 275, N, n]/Lc], {n, 26, 16, -2}, {N, 1, 6}],
   TableHeadings → {Range[26, 16, -2], Range[6]}]],
 {"AWG", "Number of Conductors Each Way", "Delivered Power (W)"}, {Left, Top, Bottom}]
```

Number of Conductors Each Way

| AWG | 1 | 2 | 3 | 4 | 5 | 6 |
|---|---|---|---|---|---|---|
| 26 | 29 | 58 | 87 | 116 | 145 | 174 |
| 24 | 46 | 92 | 139 | 185 | 231 | 277 |
| 22 | 73 | 147 | 220 | 294 | 367 | 441 |
| 20 | 117 | 233 | 350 | 467 | 584 | 700 |
| 18 | 186 | 371 | 557 | 743 | 928 | 1114 |
| 16 | 295 | 590 | 886 | 1181 | 1476 | 1771 |

Delivered Power (W)

We can see that delivering the required power of 67 W requires a cable with a minimum of six 26 AWG conductors (three each way), or four 24 AWG conductors, or two 22 AWG or larger conductors.

## Connecting Closer to the MARS Node

For equipment closer to the MARS node, users have the option of connecting to a source voltage of 48 VDC. The lower voltage can't send power as far but avoids the safety hazard of 375 VDC. Falmat XtremeNet FM022208-03 cable has two 18 AWG copper power conductors along with two-pair Cat 5E Ethernet signal wires.

**What is the maximum power that could be delivered to a science instrument's Vicor DC-DC converter connected to MARS 48 VDC with a 75 m length of FM022208-03 cable?**

Vicor's 48 VDC input converters have an allowable input range of 36-75 VDC. Increasing the minimum input voltage by 10% to 40 VDC, the cable has a system wattacity of

```
Wsvicor = Ws[48, 40, 1, 18]
```

```
7.83873
```

The lay angle of the cable isn't specified, so allowing for a 10% lengthening factor, the maximum power delivered is

$$\text{Pvicor} = \frac{\text{Wsvicor}}{0.075 \times 1.1}$$

```
95.015
```

The maximum power allowed on a 48 VDC connection to MARS is 100 W, so we need to see how much total power we're requesting in this case. The current in the cable is

$$\text{Ic} = \frac{\text{Pvicor}}{40}$$

```
2.37537
```

The power lost in the cable is the current times the voltage drop.

```
Pc = Ic (48 - 40)
```

```
19.003
```

The total power being requested from MARS is

```
Pvicor + Pc
```

```
114.018
```

In this case the power will be limited by the MARS port and not the cable. We can quickly find the resistance of the cable from our previous calculations.

$$\text{Rctotal} = \frac{48 - 40}{\text{Ic}}$$

```
3.36789
```

With the MARS port providing 100 W, the cable current is

$$\text{Iclim} = \frac{100}{48} \text{ // N}$$

```
2.08333
```

The power lost in the cable is now

$$\text{Pclim} = \text{Iclim}^2 \text{ Rctotal}$$

```
14.6176
```

The power delivered to the Vicor, limited by the MARS port, is now

```
100 - Pclim
```

```
85.3824
```

# Bibliography

American wire gauge. (2012, November 1). In Wikipedia, *The Free Encyclopedia*. Retrieved 01:31, November 7, 2012, from http://en.wikipedia.org/w/index.php?title=American_wire_gauge&oldid=520859416

Cartwright, Kenneth V. Non-Calculus derivation of the maximum power transfer theorem. *The Technology Interface Journal*, Spring, 2008, Vol. 8 No. 2, http://technologyinterface.nmsu.edu

Oberg, Erik, and Christopher J. McCauley. *Machinery's handbook*, 27th edition. 27th ed. New York: Industrial Press, 2004.